MINTEK

REPORT

No. M61

AN ATOMIC-ABSORPTION PROGRAMME FOR THE APPLE II PLUS COMPUTER

by

J.H. Wepener and D.C.G. Pearton

30th October, 1982

COUNCIL FOR MINERAL TECHNOLOGY
200 Hans Strijdom Road
RANDBURG
South Africa



MINTEK

(ANALYTICAL CHEMISTRY DIVISION)

REPORT
No. M61

AN ATOMIC-ABSORPTION PROGRAMME FOR THE APPLE II PLUS COMPUTER

by

Jienie Henrietta Wepener and Donald Charles Goss Pearton

30th October, 1982

Printed and published by the Council for Mineral Technology

ISBN 0 86999 595 2

Investigators J.H. Wepener and D.C.G. Pearton
Acting Director of Division R.C. Mallett

Programme No. 013
Project No. 00682/1
Project Report No. 2

All correspondence to Mintek, Private Bag X3015, Randburg, 2125 South Africa

SYNOPSIS

An interactive computer programme, the AA-PROGRAM APPLE, has been designed and written to process data obtained during routine analysis by atomic-absorption spectrophotometry.

The programme is fast, convenient for the user, and was found to perform satisfactorily during routine operation in the laboratory. The computer used is an Apple II Plus with a video screen, and the language of the programme is Applesoft BASIC.

Operating instructions for the computer and a printout of the programme are given in the Appendices.

SAMEVATTING

Daar is 'n wisselwerkende rekenaarprogram, die AA-PROGRAM APPLE, ontwerp en geskryf om die data wat tydens roetineontledings deur atoomabsorpsiespektrofotometrie verkry is, te verwerk.

Die program is vinnig en gerieflik om te gebruik en daar is gevind dat dit bevredigend werk tydens roetinegebruik in die laboratorium. Die rekenaar wat gebruik word, is 'n Apple II Plus met 'n videoskerm en die taal van die program is Applesoft BASIC.

Gebruiksinstruksies vir die rekenaar en 'n druksel van die program word in die Aanhangsels aangegee.

CONTENTS

1. INTRODUCTION

Before 1981, the National Institute for Metallurgy (NIM), now the Council for Mineral Technology (Mintek), used a Texas computer Model 980A to process atomic-absorption (AA) data. This computer broke down early in 1981, and a Hewlett Packard H.P.97S programmable calculator was programmed to accept the data[1]. However, the H.P.97S has a limited memory capacity and is not entirely adequate for the processing of all the data that are obtained by the use of various atomic-absorption procedures. The Analytical Chemistry Division of Mintek therefore acquired an Apple II Plus computer in February 1982 and, using the programme for the Texas computer as a starting point, designed an interactive programme to fulfil the specific needs of the Atomic Absorption Section of the Division. This programme is called the AA-PROGRAM APPLE.

2. EQUIPMENT

The equipment consists of an Apple II Plus computer with a video screen, a National Centronics 703 printer console (using paper measuring 11 by 9½ in, i.e., 28 by 24 cm), and a Philips N2226 tape recorder. The programme language is Applesoft BASIC.

3. DESCRIPTION OF THE PROGRAMME

The programme uses the term correlation index (C.I.) instead of correlation coefficient (C.C.) to indicate the accuracy of the calibration graph. C.I. is derived from C.C. by use of the formula

$$\text{C.I.} = \sqrt{\frac{1 + \text{C.C.}}{1 - \text{C.C.}}}.$$

The values derived from this formula are integers, whereas C.C. values are decimal fractions. From the point of view of the user, evaluation of the calibration graphs is much easier when C.I. figures are used, and the programming for selection of the 'worst point' (outlier) is simpler.

After calling up the programme, the user enters the data for the standards.

The programme uses all the points to calculate the C.I., with an option for the user to inspect C.I.s for each individual point rejected in turn. When the C.I. is above 20, the user is reminded visually that rejection is unnecessary. (A C.I. of 20 corresponds to a C.C. of 0,995.)

The worst point (rejection of which gives the highest C.I. figure) is indicated, with an option for the user to reject it if necessary. If the option is exercised, this is indicated on the printout so that the correctness of the decision can be checked later.

The computer calculates the averages of the absorption values and only these averages are printed. Up to 10 absorption values can be entered for a sample or a standard.

The slope (M) and the constant (C) of a calibration line are printed so that the same calibration line can be used at a later stage by the re-entering of M and C.

All the data for a set of samples are collated on a laboratory worksheet before being entered into the programme. Unlike the previous programmes, which accommodated only four dilution figures, this programme can accommodate any number of dilution factors, which do not need to be calculated separately, being applied as they are entered. The first sample dilutions are always entered fully; if the sample is not diluted, the number 1 is entered. The computer then offers the user three options, as follows.

The question, 'Any new dilutions, masses, or aliquots?' is asked. If the user answers N (for NO), the programme proceeds to the absorption values for the samples. Should the user answer Y (for YES) to the question, he has two further options, namely, to enter only a new mass or first aliquot portion and then go on to sample absorption or to enter a completely new series of dilutions.

These options facilitate routine work where many samples are diluted to the same extent or differ only in mass.

The programme incorporates calculations on recovery* and provides a corrected value for the spiked sample. Absorbance values for the spiking additions are entered directly after those for the spiked sample.

After the sample concentrations on a laboratory worksheet have been calculated by the programme, a summary of the results is printed. The same calibration is used again for all the subsequent worksheets, after which the computer returns to the start of the programme, ready for the next calibration.

As was mentioned earlier, the printout does not contain all the absorption values, but only their averages. The printout is therefore fairly short, which simplifies checking.

*In this report, *recovery* means amount determined as a percentage of the amount present.

4. CONCLUSIONS

A programme has been developed for an Apple II Plus computer that is capable of processing all the data obtained during routine analysis by atomic-absorption spectrophotometry. The programme is fast, convenient for the user, and was found to perform satisfactorily during routine operation in the laboratory.

5. REFERENCE

1. PEARTON, D.C.G. Atomic-absorption programmes for the Hewlett Packard H.P.97S programmable calculator. Randburg, Council for Mineral Technology, *Report no. M24*. 1982.

APPENDIX I

OPERATING INSTRUCTIONS FOR THE AA-PROGRAM APPLE ON THE APPLE II PLUS COMPUTER

The following are the operating instructions for the user of the Apple II Plus computer.

1. Clear the previous programme by typing NEW.
2. Load the prerecorded AA-PROGRAM APPLE from the cassette.
3. Type RUN to start the programme.
4. Follow the instructions that appear on the video screen.
5. Correct the typing errors (before pressing the RETURN button) in one of the following ways.
   (a) Use the arrow key until the flashing prompt is over the incorrect value. Retype from this point.
   (b) Press the CTRL button and, while holding it down, type X. Retype the whole line.

The programme will automatically return to the start when Q is typed. This is after the last laboratory worksheet has been processed.

APPENDIX II

COMPUTER PRINTOUT OF THE AA-PROGRAM APPLE FOR USE WITH ATOMIC-ABSORPTION SPECTROPHOTOMETRY

```
10  PR# 1: PRINT : PRINT : PR# 0
20  PRINT
30  PRINT
40  PRINT "AA-PROGRAM APPLE:82-05
    -04
50  PRINT
60 Z9 = 1000
70  DIM C(100),A(10),S$(100)
80  DIM XF(50)
90  PRINT : PRINT : PR# 0
100  PRINT "READY TO RUN"
110  PRINT
120  PR# 1
130  INPUT "ELEMENT ";E$
140  PRINT
145  INPUT "DO YOU WANT TO USE PR
    EVIOUS CAL. DATA?(Y OR N)";G
    $
146  IF G$ = "Y" THEN  GOTO 2540
150  PRINT
160  PRINT "STANDARD DATA:"
170  PRINT
175  PRINT "AVERAGE ABSORBANCES:
    "
180  PR# 0
190  INPUT "NO OF STDS (EXCL BLAN
    K)= ";N1
200  PRINT
210  PRINT "STARTING WITH BL:ENT
    CONC,THEN ABS(END WITH -1)"
220  PRINT
230  FOR J = 0 TO N1
250  PRINT "PNT.";J;
260  INPUT C
270 N = 1
280  INPUT A1
290  INPUT A2
300  IF A2 =  - 1 THEN  GOTO 340
310 A1 = A1 + A2
320 N = N + 1
330  GOTO 290
340 A1 = A1 / N
341  PR# 1
343  PRINT C;" PPM= ";A1
345  PR# 0
350  IF C = 0 THEN  GOTO 2390
360 A1 = A1 - BL
370 C(J) = Z9 / C
380 A(J) = Z9 / A1
390  NEXT J
400 NM = 0:A = 0:B = 0:C = 0:D =
    0:E = 0
```

```
410  FOR J = 1 TO N1
420  IF A(J) < 0 THEN  GOTO 490
430 A = A + C(J)
440 B = B + A(J)
450 C = C + A(J) * C(J)
460 D = D + C(J) ↑ 2
470 E = E + A(J) ↑ 2
480 NM = NM + 1
490  NEXT J
500 CM = A / NM
510 AM = B / NM
520 N2 = 0:CD = 0:AD = 0
530  FOR J = 1 TO N1
540  IF A(J) < 0 THEN  GOTO 600
550 DL =  ABS (C(J) - CM)
560 CD = CD + DL
570 DL =  ABS (A(J) - AM)
580 AD = AD + DL
590 N2 = N2 + 1
600  NEXT J
610 M = AD / CD
620  PRINT
630  PR# 1
640  PRINT "SLOPE M= ";M
642  PR# 0
650  PRINT
660  IF N2 <  > N1 GOTO 1000
670 TM = 0
680 T = (NM * C - A * B) / ( SQR
     (NM * D - A ↑ 2) *  SQR (NM *
     E - B ↑ 2))
690 T =  SQR ((1 + T) / (1 - T))
700 T =  INT (T + 0.5)
710  PR# 1
720  PRINT "C.I. FOR ALL POINTS=
     ";T
725  PR# 0
730  PRINT
740  INPUT "DO YOU WANT CI'S FOR
     REJECTED POINTS?(Y OR N) ";A
     $
750  PRINT
760  IF A$ = "N" THEN  GOTO 990
761  PR# 1
762  IF T =  > 20 THEN  PRINT "!!
      -YOU ARE AWARE THAT REJECTI
     ON IS NOT NECESSARY?"
766  PR# 0
770  FOR J = 1 TO N1
780 A1 = A - C(J)
790 B1 = B - A(J)
800 C1 = C - A(J) * C(J)
810 D1 = D - C(J) ↑ 2
820 E1 = E - A(J) ↑ 2
830 BN = N2 - 1
840  IF J = 0 THEN BN = N2
```

```
850 T = (BN * C1 - A1 * B1) / ( SQR
    (BN * D1 - A1 ↑ 2) *  SQR (B
    N * E1 - B1 ↑ 2))
860 T(J) =  SQR ((1 + T) / (1 - T
    ))
870 T(J) =  INT (T(J) + 0.5)
875  PR# 1
880  PRINT "C.I.(-PNT";J;")=";T(J
    )
885  PR# 0
890  PRINT
900  IF T(J) > TM THEN WP = J
910  IF T(J) > TM THEN TM = T(J)
920  NEXT J
930 CS = 0:NS = 0
940  PRINT "DO YOU WANT TO REJECT
     WORST POINT,WHICH IS POINT
    NO.";WP;"(Y OR N)?"
950  INPUT A$
951  PR# 1
955  IF A$ = "Y" THEN  PRINT "POI
    NT NO ";WP;" REJECTED"
956  PR# 0
960  IF A$ = "N" THEN  GOTO 990
970 A(WP) =  - A(WP)
980  GOTO 520
990 CS = 0:NS = 0
1000  FOR J = 1 TO N1
1010  IF A(J) < 0 THEN  GOTO 1050

1020 C = A(J) - M * C(J)
1030 CS = C + CS
1040 NS = NS + 1
1050  NEXT J
1060 C = CS / NS
1070  PR# 1: PRINT "CONST C= ";C:
     PR# 0
1080 N = 1
1085  PR# 0
1120  PRINT "ENT SAMPLE BL ABS(EN
    D WITH -1)"
1160  INPUT A1
1170  INPUT A2
1180  IF A2 =  - 1 THEN  GOTO 122
    0
1190 A1 = A1 + A2
1200 N = N + 1
1210  GOTO 1170
1220 A1 = A1 / N
1230 BL = A1
1235  PR# 1
1240  PRINT "AVE SAMPLE BL= ";BL
1250  PR# 0
1260  PRINT
1270  INPUT "ENT WORKSHEET NO (AF
    TER LAST WORKSHEET ON THIS C
    AL CURVE,ENT Q)";K$
```

```
1280  PR# 1
1290  PRINT
1300 H = 0
1310  IF K$ = "Q" THEN  GOTO 90
1320  PRINT "$$$$WORKSHEET NO: ";
     K$;" $$$$"
1340  PR# 0
1350  PRINT "ENT SAMPLE NO.;FOR S
     PIKE ENT SP"
1360  PRINT "(AFTER LAST SAMPLE E
     NT Z)"
1370  INPUT S$
1380  IF S$ = "SP" THEN  GOTO 192
     0
1390  IF S$ = "Z" THEN  GOTO 2430

1400 H = H + 1
1410 S$(H) = S$
1420  PRINT
1425  IF D = 0 THEN  GOTO 1440
1430  INPUT "ANY NEW DILUTIONS,AL
     IQUOTS OR WEIGHTS ?";A$
1440  PR# 1
1450  PRINT "----";S$;"----"
1470  PR# 0
1490  IF A$ = "N" THEN  GOTO 1700

1510  PRINT "ENT ALIQUOT OR WEIGH
     T"
1520  INPUT W
1521  PR# 1
1523  PRINT "DILUTIONS: ";W
1530  PR# 0
1545  IF D = 0 THEN  GOTO 1590
1550  INPUT "DO YOU WANT PREV DIL
     FACTOR?(Y OR N)-";B$
1580  IF B$ = "Y" THEN  GOTO 1700

1590  PRINT "ENT VOLS & ALQS (END
      WITH -1)"
1600 N = 1
1610  PR# 1
1620  INPUT D
1630  INPUT C1
1640  IF C1 =  - 1 THEN  GOTO 170
     0
1650 N = N + 1
1660  IF  INT (N / 2) = N / 2 THEN
     C1 = 1 / C1
1670 D = D * C1
1680  GOTO 1630
1700  PR# 0
1710  PRINT "ENT SAMPLE ABS (END
     WITH -1)"
1720 N = 1
1730  INPUT A1
1740  INPUT A2
```

```
1750  IF A2 =  - 1 THEN  GOTO 179
     0
1760 A1 = A1 + A2
1770 N = N + 1
1780  GOTO 1740
1790 A1 = A1 / N
1794  PR# 1
1795  PRINT "AVE ABS= ";A1
1796  PR# 0
1800 A1 = A1 - BL
1810 A1 = Z9 / A1
1820 X = (A1 - C) / M
1830 X = Z9 / X
1840 F = D / W
1860 XF(H) = X * F
1870 XF(H) =  INT (XF(H) * 1000 +
     0.5) / 1000
1880  PR# 1
1890  PRINT "****CONC OF SAMPLE "
     ;S$;" = ";XF(H);" ****"
1895  PRINT
1900  PR# 0
1910  GOTO 1340
1920  PRINT
1930  PR# 1
1940  PRINT "SPIKE: "
1945  PR# 0
1960  PRINT "ENT SAMPLE ABS (END
     WITH -1)"
1965 N = 1
1980  INPUT S1
1990  INPUT S2
2000  IF S2 =  - 1 THEN  GOTO 2040
     0
2010 S1 = S1 + S2
2020 N = N + 1
2030  GOTO 1990
2040 S1 = S1 / N
2042  PR# 1
2045  PRINT "AVE SAMPLE ABS= ";S1

2047  PR# 0
2050 X3 = (S1 - C) / M
2060 C4 = X3
2065  PR# 0
2070  PRINT "ENT (SAMPLE+SP)ABS (
     END WITH -1)"
2080 N = 1
2090  INPUT S1
2100  INPUT S2
2110  IF S2 =  - 1 THEN  GOTO 215
     0
2120 S1 = S1 + S2
2130 N = N + 1
2140  GOTO 2100
2150 S1 = S1 / N
2152  PR# 1
```

```
2154  PRINT "AVE(SAMPLE+SP)ABS= "
     ;S1
2156  PR# 0
2160 X1 = (S1 - C) / M
2170 C2 = X1
2175  PR# 0
2180  PRINT "ENT (SP ONLY)ABS (EN
     D WITH -1)"
2190 N = 1
2200  INPUT S1
2210  INPUT S2
2220  IF S2 =  - 1 THEN  GOTO 226
     0
2230 S1 = S1 + S2
2240 N = N + 1
2250  GOTO 2210
2260 S1 = S1 / N
2262  PR# 1
2264  PRINT "AVE(SP ONLY)ABS= ";S
     1
2266  PR# 0
2270 X2 = (S1 - C) / M
2280 C3 = X2
2290 R = ((C2 - C4) / C3) * 100
2300 R =  INT (R + 0.5)
2310  PR# 1
2320  PRINT "% RECOVERY= ";R
2330 T = (100 * XF(H)) / R
2340 T =  INT (T * 1000 + 0.5) /
     1000
235'   'RINT "CORRECTED VALUE= ";T

236    R# 0
237    RINT
2380  GOTO 1340
2390 BL = A1
2400  GOTO 390
2410  PRINT
2420  PRINT
2430  PRINT "SUMMARY OF RESULTS"
2440  PR# 1
2445  PRINT
2450  PRINT "AA RESULTS FOR ";E$;
     " ON WORKSHEET ";K$
2460  PRINT
2470  FOR J = 1 TO H
2480  PRINT "SAMPLE NO ";S$(J);"-
     ---";XF(J);" PPM"
2490  NEXT J
2500  PRINT
2510  PRINT
2520  PR# 0
2530  GOTO 1270
2540  INPUT "RE-ENTER SLOPE M-";M

2545  INPUT "RE-ENTER CONST.C-";C

2550  GOTO 1080
```